Formalizing higher categories with simplicial homotopy type theory

Thesis defense talk

Theofanis Chatzidiamantis-Christoforidis
University of Bonn

July 29, 2024

This is the third (and final) talk in a series aimed at introducing simplicial type theory, following the work of Riehl and Shulman [RS17], and to present formalization work in the Rzk proof assistant. We also give a brief overview of the semantics of simplicial type theory in Reedy fibrant simplicial spaces, working towards an experimental project for the formalization of properties of 2-Segal spaces.

Contents

1 Constructing a type theory that can reason about (complete) Segal spaces

Recollections on Segal spaces

We spend some more time talking about Segal spaces, as the type theory we will present relies on a modified version of the Segal condition, which still has Segal spaces as a model for what we will call Segal types, and in complete Segal spaces for Rezk types, which are again Segal types with an extra completeness condition.

Recall that we denote by $s\mathbf{Set} := \mathrm{Fun}(\Delta^{op}, \mathbf{Set})$ the category of *simplicial sets* and by $s\mathcal{S} := \mathrm{Fun}(\Delta^{op}, s\mathbf{Set})$ the category of *simplicial spaces*. In particular, we are concerned with *Reedy fibrant simplicial spaces*, a condition which guarantees:

- Every level W_n of a Reedy fibrant simplicial space W is a Kan complex. This means that we have *spaces* of objects, morphisms, 2-cells, etc.
- The source-target map $W_1 \xrightarrow{(d_1,d_0)} W_0 \times W_0$ is a Kan fibration.

1.1 Definition. Let W be a Reedy fibrant simplicial space. W is called a *Segal space* if the Segal maps

$$W_n \xrightarrow{\varphi_n} W_1 \times_{W_0} \cdots \times_{W_0} W_1$$

are weak equivalences for all $n \geq 2$.

To say that an extension of homotopy type theory has a model in Segal spaces, we first need to make sure that we can interpret HoTT itself in such a model.

1.2 Theorem ([Shu15, Theorem 6.4]). *The category $s\mathcal{S}$ of simplicial spaces with the Reedy model structure (actually, every category of the form* $\mathrm{Fun}(\mathcal{C}^{op}, s\mathbf{Set})$ *where $\mathcal{C}$ is an elegant Reedy category) supports a model of intensional type theory with dependent sums and products, identity types, and as many univalent universes as inaccessible cardinals (greater than $|\mathcal{C}|$).*

We will now focus on a challenge that will come up again later: How do we reason about Segal spaces in a type theory (and a proof assistant) without a natural numbers type? We need to reduce the Segal condition to a finite number of coherence conditions.

A closer look at $s\mathcal{S}$

There are two ways that we can think of a simplicial set X as a simplicial space:

- The ***vertical embedding*** $i_F(X)_{nm} := X_n$, and
- The ***horizontal embedding*** $i_\Delta(X)_{nm} := X_m$

1.3 Notation. For $n \in \mathbb{N}$ we define $F(n) := i_F(\Delta^n)$ and $\Delta[n] := i_\Delta(\Delta^n)$.
We also set $\partial F(n) := i_F(\partial\Delta^n)$ and $L(n)_k := i_F(\Lambda^n_k)$.

$s\mathcal{S}$ is cartesian closed, with the exponential being the ***mapping simplicial space***

$$(W^Z)_{nm} = \mathrm{Hom}_{s\mathcal{S}}(F(n) \times \Delta[m] \times Z, W)$$

We define its 0-th level to be the *mapping simplicial set*

$$\mathrm{Map}_{s\mathcal{S}}(Z, W) = (W^Z)_0 \cong \mathrm{Hom}_{s\mathcal{S}}(\Delta[\bullet] \times Z, W)$$

We can now apply the Yoneda lemma to this mapping space and get an isomorphism of simplicial sets

$$\mathrm{Map}_{s\mathcal{S}}(F(n), W) \cong W_n$$

1.4 Proposition (Reedy model structure). *There is a model structure on $s\mathcal{S}$ such that*

- *The fibrations are the maps $f : X \to Y$ for which the morphism of simplicial sets*

$$\mathrm{Map}_{s\mathcal{S}}(F(n), X) \to \mathrm{Map}_{s\mathcal{S}}(\partial F(n), X) \times_{\mathrm{Map}_{s\mathcal{S}}(\partial F(n), Y)} \mathrm{Map}_{s\mathcal{S}}(F(n), Y)$$

  is a Kan fibration for all $n \geq 0$.

- *The weak equivalences are the maps that are levelwise weak equivalences in $s\mathcal{S}\mathrm{et}_{Quillen}$.*

1.5 Proposition ([BR12, 3.15+4.5]). *The Reedy and injective model structures on $s\mathcal{S}$ coincide. Equivalently, the cofibrations in $s\mathcal{S}_{Reedy}$ are precisely the monomorphisms.*

Unfolding the definition, we get that a simplicial space X is *Reedy fibrant* if the map

$$\mathrm{Map}_{s\mathcal{S}}(F(n), X) \to \mathrm{Map}_{s\mathcal{S}}(\partial F(n), X)$$

is a Kan fibration for all $n \geq 0$.

Thinking about **The Joyal-Tierney calculus** ruined my day

1.6 Definition ([JT07, Section 2]). Let X, Y be two simplicial sets. The ***box product*** $X \square Y$ is the *simplicial space* obtained by setting

$$(X \square Y)_{mn} \coloneqq X_m \times Y_n$$

In particular, we have $i_F(X) \cong X \square \Delta^0$.

1.7 Notation. Let $\mathcal{A}, \mathcal{B}$ be sets of morphisms in a category $\mathcal{C}$. We write $\mathcal{A} \pitchfork \mathcal{B}$ if $\mathcal{A}$ has the *left lifting property (LLP)* against $\mathcal{B}$, i.e., there exist lifts for all commutative squares

$$\begin{tikzcd} A \arrow[r] \arrow[d, "\mathcal{A} \ni f"'] & B \arrow[d, "g \in \mathcal{B}"] \\ A' \arrow[r] \arrow[ur, dotted] & B' \end{tikzcd}$$

We also write $f \pitchfork g$ for individual morphisms f, g such that f has the LLP against g. We denote by ${}^{\pitchfork}\mathcal{A}$ (resp. $\mathcal{A}^{\pitchfork}$) the class of morphisms having the left (resp. right) lifting property against $\mathcal{A}$.

1.8 Definition. Let $\mathcal{C}$ be a category with small (co)limits.

1. Let $f : A \to B, g : C \to D$ be morphisms in $\mathcal{C}$. The ***pushout product***

$$f \square g : A \times D \coprod_{A \times C} B \times C \to B \times D$$

is the induced morphism

$$\begin{tikzcd}
A \times C \arrow[r, "f\times id_C"] \arrow[d, "id_A\times g"'] & B \times C \arrow[d] \arrow[ddr, bend left, "id_B\times g"] & \\
A \times D \arrow[r] \arrow[drr, bend right, "f\times id_D"'] & A \times D \coprod_{A\times C} B \times C \arrow[dr, dotted, "f\Box g"] & \\
& & B \times D
\end{tikzcd}$$

2. Assume that $\mathcal{C}$ is also cartesian closed. Let $g : C \to D$ and $u : X \to Y$ be morphisms in $\mathcal{C}$. We then have the ***pullback exponential***

$$\exp(g, u) : D^X \to Y^D \times_{X^C} X^D$$

induced by the pullback

$$\begin{tikzcd}
X^D \arrow[dr, dotted, "\exp(g,u)"] \arrow[drr, bend left, "u^D"] \arrow[ddr, bend right, "X^g"'] & & \\
& Y^D \times_{X^C} X^D \arrow[r] \arrow[d] & Y^D \arrow[d, "Y^u"] \\
& X^C \arrow[r, "u^C"'] & Y^C
\end{tikzcd}$$

Using the adjunctions provided by the cartesian closedness of $\mathcal{C}$, we can prove that we can interchange between lifting problems:

1.9 Proposition ([JT07, Proposition 7.6]). *Let $\mathcal{C}$ be a cartesian closed category with small (co)limits and $f : A \to B$, $g : C \to D$, $u : X \to Y$ be morphisms in $\mathcal{C}$. Then*

$$f\Box g \pitchfork u \Leftrightarrow f \pitchfork \exp(g, u) \Leftrightarrow g \pitchfork \exp(f, u)$$

This means that we can reduce solving lifting problems to checking for certain generating morphisms, by interchanging between fibrations and cofibrations with the proposition above.

Using the Joyal-Tierney calculus, Riehl and Shulman were able to prove:

1.10 Theorem ([RS17, A.21]). *A Reedy fibrant simplicial space W is Segal if and only if the map*

$$W^{F(2)} \to W^{L(2)_1}$$

is a trivial Reedy fibration.

Of course, the hom-types that simplicial type theory adds will themselves be types! We have, again, an analogue in simplicial spaces:

1.11 Definition. Let W be a simplicial space and $x, y : F(0) \to W$. The ***mapping simplicial space*** $\mathrm{Hom}_W(x, y)$ is defined as the pullback of simplicial spaces

$$\begin{tikzcd} \mathrm{Hom}_W(x,y) \arrow[r] \arrow[d] & W^{F(1)} \arrow[d] \\ F(0) \arrow[r, "(x,y)"] & W \times W \end{tikzcd}$$

Cubes, topes, and shapes in type theory

We now introduce a non-fibrant layer to our type theory, which will allow us to define our simplices and horns. We consider shapes that can be "carved out" of cubes using simple logical formulas. We have:

- A layer of ***cubes***: Cubes are generated by the one-element type $\mathbf{1}$ and the ***directed interval*** $\mathbb{2}$ plus a theory of ***finite products*** $\times$.
- A layer of ***topes***: These are the conditions that we can impose on terms of cubes. The formulas we allow are made up of a simple intuitionistic logic, generated by judgmental equality $\equiv$, conjunction $\vee$, disjunction $\wedge$ and an "inequality tope"
$$x, y : \mathbb{2} \vdash (x \leq y) \ \mathsf{tope}$$
- A basic theory for cubes: We postulate that $x \equiv \star$ for any $x : \mathbf{1}$ and that $\mathbb{2}$ has two endpoints $0_{\mathbb{2}}, 1_{\mathbb{2}} : \mathbb{2}$. To make $\mathbb{2}$ directed, we impose the relations
$$x, y : \mathbb{2} \vdash x \leq y \vee y \leq x\,, \quad x : \mathbb{2} \vdash 0_{\mathbb{2}} \leq x \leq 1_{\mathbb{2}}$$
in addition to axioms that $\leq$ defines an ordering on terms of $\mathbb{2}$. We also say that a tuple $\langle t_1, t_2 \rangle : I \times J$ for I, J cube, is fully determined by its projections and they work as expected.

Making shapes. We can now consider cubes restricted by topes. We add a shape-forming axiom:
$$\frac{I \ \mathsf{cube} \quad t : I \vdash \phi \ \mathsf{tope}}{\{t : I \mid \phi\} \ \mathsf{shape}}$$

The relations above are now enough to define simplices.

1.12 Definition. We define the ***n-simplex***

$$\Delta^n :\equiv \{\langle t_1, \dots, t_n \rangle : \mathbb{2}^n \mid t_n \leq \dots \leq t_1\}$$

1.13 Example. Noticing that the condition on the right always holds for the first two simplices, we can simplify to $\Delta^0 = \{t : \mathbf{1} \mid \top\}$ and $\Delta^1 = \{t : \mathbb{2} \mid \top\}$, i.e., the point and the interval.

Adding more conditions to our topes, we can now define basic boundaries and horns as subshapes of our simplices:

- $\partial\Delta^1 \coloneqq \{t : \mathbb{2} \mid t \equiv 0 \vee t \equiv 1\}$ ("endpoints of $\mathbb{2}$")
- $\partial\Delta^2 \coloneqq \{\langle t_1, t_2\rangle : \mathbb{2} \times \mathbb{2} \mid (0 \equiv t_2) \vee (t_1 \equiv t_2) \vee (t_1 \equiv 1)\}$ (boundary defined by two sides and the diagonal)
- $\Lambda^2_1 \coloneqq \{\langle t_1, t_2\rangle : \mathbb{2} \times \mathbb{2} \mid (0 \equiv t_2) \vee (t_1 \equiv 1)\}$

1.14 Remark. Note that we did not define horns and boundaries in general. However, in practice we will only need low-dimensional simplices and boundaries, usually of dimension at most 3.

Mapping out of shapes by $\vee$-recursion. For shapes $\{t : I \mid \phi\}$, we introduce an additional dependent product type

$$\prod_{t:I|\phi} A(t)$$

Such a type will have its formation rules, which are recursive on $\vee$:

To construct a dependent function $f : \prod_{t:I|\phi\vee\psi} A(t)$, all we have to do is to provide functions out of $\{t : I \mid \phi\}$ and $\{t : I \mid \psi\}$ which agree when $\phi \wedge \psi$ holds.

Extension types: lifting along shape inclusions

As is the case with all types we used before constructing shapes, one can define (dependent) function types where the base type is a shape. Since the goal here is to express equivalents of filling conditions, a theory of *liftings* is needed. Dependent function types out of shapes therefore come with the more general notion of an extension type along a subspace inclusion.

1.15 Notation. For a given cube I, we write $\{t : I \mid \phi\} \subseteq \{t : I \mid \psi\}$ if $t : I \mid \phi \vdash \psi$.

Then, for a shape inclusion $\{t : I \mid \phi\} \subseteq \{t : I \mid \psi\}$, we can introduce the extension type

$$\frac{\begin{array}{c}\{t : I \mid \phi\}\ \mathsf{shape} \quad \{t : I \mid \psi\}\ \mathsf{shape} \quad t : I \mid \phi \vdash \psi \\ \Xi|\Phi \vdash \Gamma\ \mathsf{ctx} \quad \Xi, t : I \mid \Phi, \phi \mid \Gamma \vdash A\ \mathsf{type} \quad \Xi, t : I \mid \Phi, \phi \mid \Gamma \vdash a : A\end{array}}{\Xi|\Phi|\Gamma \vdash \left\langle \prod_{t:I|\psi} A(t)\middle|^{\phi}_{a} \right\rangle\ \mathsf{type}}$$

We think of an extension type $\langle \prod_{t:I|\psi} A|^{\phi}_{a}\rangle$ as the type of dependent functions that, when restricted to the subshape $\{t : I \mid \phi\}$ (i.e., when ϕ holds), map to a. Informally (and for a constant family A) what we want is a type of lifting diagrams

$$\begin{tikzcd} \{t : I \mid \phi\} \arrow[r] \arrow[d, hook] & A \\ \{t : I \mid \psi\} \arrow[ur, dashed] & \end{tikzcd}$$

The rest of the type-forming rules for extension types are the same as the usual dependent product types, adding the restriction condition (judgementally) for elimination rules. In formal terms:

$$\frac{\{t : I \mid \phi\} \text{ shape} \quad \{t : I \mid \psi\} \text{ shape} \quad t : I \mid \phi \vdash \psi \qquad \Xi \mid \Phi \mid \Gamma \vdash f : \langle \prod_{t:I|\psi} A(t)|^{\phi}_{a} \rangle \quad \Xi \vdash s : I \quad \Xi \mid \Phi \vdash \phi(s)}{\Xi \mid \Phi \mid \Gamma \vdash f(s) \equiv a[s/t]}$$

1.16 Theorem ("Axiom" of choice, [RS17, 4.2]). *For a shape inclusion $\{t : I \mid \phi\} \subseteq \{t : I \mid \psi\}$, a family $X : \{t : I \mid \psi\} \to \mathcal{U}$, $Y : \prod_{t:I|\psi}(X \to \mathcal{U})$, and dependent functions $a : \prod_{t:I|\phi} X(t)$, $b : \prod_{t:I|\phi} Y(t, a(t))$, there is an equivalence*

$$\left\langle \prod_{t:I|\psi} \Big(\sum_{x:X(t)} Y(t,x) \Big) \Big|^{\phi}_{\lambda t.(a(t),b(t))} \right\rangle \simeq \sum_{f:\langle \prod_{t:I|\psi} X(t)|^{\phi}_{a} \rangle} \left\langle \prod_{t:I|\psi} Y(t, f(t))|^{\phi}_{b} \right\rangle$$

Proof. We can explicitly define inverse equivalences

$$\lambda g. \big(\lambda t.pr_1(g(t)),\ \lambda s.pr_2(g(s))\big)$$

$$\lambda h.\lambda H. \,(\lambda t.h(t),\ \lambda s.H(s))$$

and check that they compose precisely to identities. □

Segal and Rezk types

1.17 Definition. For a type A and $x, y : A$, we define the type of ***arrows***

$$\mathsf{hom}_A(x,y) :\equiv \left\langle \Delta^1 \to A \,|^{\partial\Delta^1}_{[x,y]} \right\rangle$$

Having a hom-type, the Segal condition is now expressed in terms of contractibility of types consisting of 2-cells witnessing candidate compositions of two composable maps. We first define the type of such 2-cells with boundaries three arrows $f : \mathsf{hom}_A(x,y)$, $g : \mathsf{hom}_A(y,z)$, $h : \mathsf{hom}_A(x,z)$ as

$$\mathsf{hom}^2_A(f,g;h) :\equiv \left\langle \Delta^2 \to A \,|^{\partial\Delta^2}_{[x,y,z,f,g,h]} \right\rangle$$

Here h is a candidate composite of f and g. Note that h is the precisely the result of evaluating at the diagonal: For $\alpha : \mathsf{hom}^2_A(f,g;h)$ and $t : \Delta^1$, we have $h(t) \equiv \alpha(t,t)$.

1.18 Definition. A is a ***Segal type*** if the type

$$\sum_{h:\mathsf{hom}_A(x,z)} \mathsf{hom}^2_A(f,g;h)$$

is contractible for all $x, y, z : A$, $f : \mathsf{hom}_A(x,y)$, $g : \mathsf{hom}_A(y,z)$.

This is precisely the condition that a composite exists and that any two candidate composites are equal.

1.19 Proposition. *If A is a Segal type, compositions are propositionally unique and the composition operation is associative.*

Isomorphisms in hom-types can be characterized in the same way that equivalences are in homotopy type theory.

1.20 Definition.

1. For $f : \mathsf{hom}_A(x, y)$ define the type

$$\mathsf{isiso}(f) := \Big(\sum_{g : \mathsf{hom}_A(y,x)} g \circ f = id_x \Big) \times \Big(\sum_{h : \mathsf{hom}_A(y,x)} f \circ h = id_y \Big)$$

   f is an ***isomorphism*** if $\mathsf{isiso}(f)$ is inhabited.

2. For fixed $x, y : A$, denote by

$$(x \cong_A y) := \sum_{f : \mathsf{hom}_A(x,y)} \mathsf{isiso}(f)$$

   the type of all isomorphisms between x and y.

1.21 Construction. By path induction, it is easy to convert paths into arrows: We can define the map

$$\mathsf{idtoarr}_A : \prod_{x,y:A} \Big((x =_A y) \to (\mathsf{hom}_A(x, y)) \Big), \quad \mathsf{idtoiso}_A(a, a, \mathsf{refl}_a) :\equiv \mathsf{id}_a$$

and, since we are mapping to identities, we can specialize to isomorphisms:

$$\mathsf{idtoiso}_A : \prod_{x,y:A} \Big((x =_A y) \to (x \cong_A y) \Big), \quad \mathsf{idtoiso}_A(a, a, \mathsf{refl}_a) :\equiv (\mathsf{id}_a, ((\mathsf{id}_a, \mathsf{refl}), (\mathsf{id}_a, \mathsf{refl})))$$

This map now allows us to express the Rezk completeness condition type-theoretically.

1.22 Definition. A Segal type A is ***Rezk*** if the map $\mathsf{idtoiso}_A(x, y, -)$ is an equivalence for all $x, y : A$.

2 Formalization of (simplicial) homotopy type theory in Rzk

Introduction

We now give some examples of code in Rzk, a proof assistant developed by Nikolai Kudasov for formalization for simplicial type theory. After some basic examples, we present some formal proofs contributed by the author as part of this thesis project, both in the homotopy type theory and simplicial type theory Rzk libraries.

2.1 Remark. As an experimental theory and proof assistant, Rzk has limitations: There is only one universe and no natural numbers type. Thus, many aspects of homotopy type theory (especially those dependent on univalence) are not accessible yet. A theoretical basis for extending simplicial type theory with univalence has recently been proposed in [GWB24], where one can actually construct universes that are Rezk types and satisfy a directed version of univalence (think of the ∞-category of spaces/anima).

2.2 Example. Here are some definitions of simplices, boundaries and horns in Rzk:

```
#def Δ¹
  : 2 → TOPE
  := \ t → TOP

#def Δ²
  : ( 2 × 2) → TOPE
  := \ (t , s) → s ≤ t

#def Δ³
  : ( 2 × 2 × 2) → TOPE
  := \ ((t1 , t2) , t3) → t3 ≤ t2 ∧ t2 ≤ t1

#def ∂Δ²
  : Δ² → TOPE
  :=
    \ (t , s) → (s ≡ 0₂ ∨ t ≡ 1₂ ∨ s ≡ t)

#def Λ
  : ( 2 × 2) → TOPE
  := \ (t , s) → (s ≡ 0₂ ∨ t ≡ 1₂)
```

Let us briefly explain the notation:

Every construction is proceeded by #def before it is given a name. The character ":" is followed by the type that we want to construct the term of. The specific term we are constructing comes after ":=".

A shape $\{t : I \mid \phi\}$ is formally written as a function $\lambda t.\phi(t) : I \to$ TOPE. λ-abstraction is denoted by \, and we do not yet use the $\mapsto$ symbol, as this is used for extension types.

2.3 Example. We now show how we can define hom-types as an example for extension type notation. We have the type of arrows with two fixed endpoints

```
#def hom
  ( A : U)
  ( x y : A)
  : U
  :=
    ( t : Δ¹)
  → A [ t ≡ 0₂ |→ x , -- the left endpoint is exactly x
        t ≡ 1₂ |→ y]  -- the right endpoint is exactly y
```

and the type of 2-cells:

```
#def hom2
  ( A : U)
  ( x y z : A)
  ( f : hom A x y)
```

```
  ( g : hom A y z)
  ( h : hom A x z)
  : U
  :=
    ( ( t₁ , t₂) : Δ²)
  → A [ t₂ ≡ 0₂ |→ f t₁ , -- the top edge is exactly `f`,
        t₁ ≡ 1₂ |→ g t₂ , -- the right edge is exactly `g`, and
        t₂ ≡ t₁ |→ h t₂]  -- the diagonal is exactly `h`
```

Here, we have started to introduce extra hypotheses before the type declaration ":", with the type now being the generic universe U. For extension types, we use $[\phi(t) \mapsto ft]$ (with my best approximation of ↦, I couldn't find a font with that amount of unicode support with the markdown package) to denote that functions need to restrict to $f(t)$ when $\phi(t)$ holds.

Note that there is no λ-abstraction now, as the notation $(x : A) \to C(x)$ is used for product types $\prod_{x:A} C(x)$.

Some contributed formal proofs

Contractible types. We prove a standard homotopy type theory statement: For a type family $C : A \to \mathcal{U}$, the total type $\sum_{x:A} C(x)$ is equivalent to the fiber $C(a_0)$, evaluated at the center of contraction $a_0 : A$ (and hence, by transport, all fibers $C(x)$).

For space-saving reasons, we include the definition of one of the two inverse equivalences (the nontrivial one), and a proof that it has a retraction (click here for the whole thing).

```
#def center-fiber-total-type-is-contr-base
  ( A : U)
  ( is-contr-A : is-contr A)
  ( C : A → U)
  : ( Σ ( x : A) , C x)
  → ( C (center-contraction A is-contr-A))
  :=
    \ (x , u) →
      transport
        ( A)
        ( C)
        ( x)
        ( center-contraction A is-contr-A)
        ( rev
          ( A)
          ( center-contraction A is-contr-A)
          ( x)
          ( homotopy-contraction A is-contr-A x))
        ( u)

#def has-retraction-center-fiber-total-type-is-contr-base
  ( A : U)
  ( is-contr-A : is-contr A)
  ( C : A → U)
  : has-retraction
    ( Σ ( x : A) , C x)
    ( C (center-contraction A is-contr-A))
    ( center-fiber-total-type-is-contr-base A is-contr-A C)
```

```
:=
  ( ( total-type-center-fiber-is-contr-base A is-contr-A C)
  , \ (x , u) →
      ( rev
        ( Σ ( x : A) , C x)
        ( x , u)
        ( total-type-center-fiber-is-contr-base A is-contr-A C
          ( center-fiber-total-type-is-contr-base A is-contr-A C (x , u)))
        ( transport-lift A C
            ( x)
            ( center-contraction A is-contr-A)
            ( rev
              ( A)
              ( center-contraction A is-contr-A)
              ( x)
              ( homotopy-contraction A is-contr-A x))
            ( u))))
```

Composing dependent arrows in covariant families. Given a type family $C : A \to \mathcal{U}$, $u : C(x)$, $v : C(y)$ and $f : \mathsf{hom}_A(x, y)$, consider the type of *dependent arrows*

$$\mathsf{dhom}_{C(f)}(u, v) := \left\langle \prod_{t:2} C(f(t))\Big|^{\partial\Delta^1}_{[u,v]} \right\rangle$$

We can easily show that there is an equivalence

$$\mathsf{hom}_{\sum_{x:A} C(x)}((x, u), (y, v)) \simeq \sum_{f:\mathsf{hom}_A(x,y)} \mathsf{dhom}_{C(f)}(u, v)$$

i.e., arrows in the total type are arrows in the base together with dependent arrows over them.

2.4 Definition ([RS17, 8.2]). A type family $C : A \to \mathcal{U}$ is ***covariant*** if for every $f : \mathsf{hom}_A(x, y)$ and $u : C(x)$, the type

$$\sum_{v:C(y)} \mathsf{dhom}_{C(f)}(u, v)$$

is contractible. Dually, C is ***contravariant*** if for every $f : \mathsf{hom}_A(x, y)$ and $v : C(y)$ the type

$$\sum_{u:C(x)} \mathsf{dhom}_{C(f)}(u, v)$$

is contractible.

2.5 Theorem. *If A is Segal and $C : A \to \mathcal{U}$ is covariant, then the total type $\sum_{x:A} C(x)$ is Segal.*

We provide a formal proof in Rzk that if $C : A \to \mathcal{U}$ is covariant, we can compose dependent arrows just like regular arrows, using analogous dependent horn constructions to prove a contractibility condition for dependent 2-cells.

The outline for this formalization was given by Emily Riehl.

As an example, we include the final steps of the proof (click here for the whole thing):

```
#def is-contr-dhom2-comp-is-covariant-family-is-segal-base uses (extext)
  ( A : U)
  ( is-segal-A : is-segal A)
  ( C : A → U)
  ( is-covariant-C : is-covariant A C)
  ( x y z : A)
  ( f : hom A x y)
  ( g : hom A y z)
  ( u : C x)
  ( v : C y)
  ( w : C z)
  ( ff : dhom A x y f C u v)
  ( gg : dhom A y z g C v w)
  : is-contr
    ( Σ ( hh : dhom A x z (comp-is-segal A is-segal-A x y z f g) C u w)
      , dhom2 A x y z f g (comp-is-segal A is-segal-A x y z f g)
        ( witness-comp-is-segal A is-segal-A x y z f g) C u v w ff gg hh)
  :=
    is-contr-equiv-is-contr'
      ( Σ ( hh : dhom A x z (comp-is-segal A is-segal-A x y z f g) C u w)
        , dhom2 A x y z f g (comp-is-segal A is-segal-A x y z f g)
          ( witness-comp-is-segal A is-segal-A x y z f g) C u v w ff gg hh)
      ( ( t : Δ²) → C ((witness-comp-is-segal A is-segal-A x y z f g) t)
                  [Λ t |→ dhorn A x y z f g C u v w ff gg t])
      ( dcompositions-are-dhorn-fillings A x y z f g
        ( comp-is-segal A is-segal-A x y z f g)
        ( witness-comp-is-segal A is-segal-A x y z f g)
          C u v w ff gg)
      ( is-contr-comp-horn-ext-is-covariant-family-is-segal-base
        ( A)
        ( is-segal-A)
        ( C)
        ( is-covariant-C)
        ( horn A x y z f g)
        ( dhorn A x y z f g C u v w ff gg))

#def dcomp-is-covariant-family-is-segal-base uses (extext)
  ( A : U)
  ( is-segal-A : is-segal A)
  ( C : A → U)
  ( is-covariant-C : is-covariant A C)
  ( x y z : A)
  ( f : hom A x y)
  ( g : hom A y z)
  ( u : C x)
  ( v : C y)
  ( w : C z)
  ( ff : dhom A x y f C u v)
  ( gg : dhom A y z g C v w)
  : dhom A x z (comp-is-segal A is-segal-A x y z f g) C u w
```

```
  :=
    ( first
      ( first
        ( is-contr-dhom2-comp-is-covariant-family-is-segal-base
          A is-segal-A C is-covariant-C x y z f g u v w ff gg)))
```

3 Future work: Formalizing 2-Segal spaces

2-Segal spaces are a generalization of Segal spaces, defined by Dyckerhoff and Kapranov [DK19]. In terms of category theory, while a (1-)Segal space is a "pre-category with compositions", a 2-Segal space has no condition on the existence or the uniqueness of compositions, but still requires higher coherence conditions to be satisfied. Since 2-Segal spaces still have associativity, they show up in many algebraic constructions, and the combination of algebraic and homotopical information is of particular use to the field of *higher algebra*.

Our aim is to use the work of Dyckerhoff-Kapranov and Feller [Fel23] to introduce a meaningful definition of a *2-Segal type* in simplicial type theory.

3.1 Proposition ([Fel23, 5.5]). *The class of 2-Segal anodyne maps in s*$\mathcal{S}$et *is generated by the set of maps*

$$\left\{(\Lambda^3_{0,2} \hookrightarrow \Delta^3)\square(\partial\Delta^n \hookrightarrow \Delta^n)\right\}_{n\geq 0} \cup \left\{(\Lambda^3_{1,3} \hookrightarrow \Delta^3)\square(\partial\Delta^m \hookrightarrow \Delta^m)\right\}_{m\geq 0}$$

Using the box product and the Joyal-Tierney calculus, we can translate that result to simplicial spaces. In particular, requiring fillings for generalized 3-horns should give the proper generalization of a Segal type corresponding to a Reedy fibrant 2-Segal space.

We show our proposed definitions in samples of Rzk code:

3.1.1 The 3 dimensional 2-Segal horns

```
#def Λ³₍₀₂₎
  : Δ³ → TOPE
  :=
    \ ((t1 , t2) , t3) → t3 ≡ 0₂ ∨ t1 ≡ t2

#def Λ³₍₁₃₎
  : Δ³ → TOPE
  :=
    \ ((t1 , t2) , t3) → t2 ≡ t3 ∨ t1 ≡ 1₂
```

3.1.2 2-Segal types

Here we define 2-Segal types with a contractibility condition.

```
#def is-2-segal₍₀₂₎
  ( A : U)
  : U
  :=
    ( w : A) → (x : A) → (y : A) → (z : A)
  → ( f : hom A w x) → (gf : hom A w y) → (hgf : hom A w z)
```

```
  → ( g : hom A x y) → (h : hom A y z)
  → ( α₃ : hom2 A w x y f g gf) → (α₁ : hom2 A w y z gf h hgf)
  → is-contr
      ( Σ ( hg : hom A x z)
      , ( Σ ( α₂ : hom2 A w x z f hg hgf)
        , ( Σ ( α₀ : hom2 A x y z g h hg)
            , hom3 A w x y z f gf hgf g hg h α₃ α₂ α₁ α₀)))

#def is-2-segal₍₁₃₎
  ( A : U)
  : U
  :=
    ( w : A) → (x : A) → (y : A) → (z : A)
  → ( f : hom A w x) → (hgf : hom A w z)
  → ( g : hom A x y) → (hg : hom A x z) → (h : hom A y z)
  → ( α₂ : hom2 A w x z f hg hgf) → (α₀ : hom2 A x y z g h hg)
  → is-contr
      ( Σ ( gf : hom A w y)
      , ( Σ ( α₃ : hom2 A w x y f g gf)
        , ( Σ ( α₁ : hom2 A w y z gf h hgf)
            , hom3 A w x y z f gf hgf g hg h α₃ α₂ α₁ α₀)))

#def is-2-segal
  ( A : U)
  : U
  :=
    product (is-2-segal₍₀₂₎ A) (is-2-segal₍₁₃₎ A)
```

Our main goal for now is to prove that this is equivalent to the type being local with respect to 2-Segal horn inclusions.

```
#def is-local-2-segal-horn-inclusion
  ( A : U)
  : U
  :=
    product
     ( is-local-type (2 × 2 × 2) Δ³ Λ³₍₀₂₎ A)
     ( is-local-type (2 × 2 × 2) Δ³ Λ³₍₁₃₎ A)
```

References

[BR12] Julia E. Bergner and Charles Rezk. *Reedy categories and the Θ-construction*. 2012. arXiv: 1110.1066v2 [math.AT]. URL: https://arxiv.org/abs/1110.1066.

[DK19] Tobias Dyckerhoff and Mikhail Kapranov. Higher Segal Spaces. Lecture Notes in Mathematics. Springer International Publishing, 2019. ISBN: 9783030271244. URL: https://doi.org/10.1007/978-3-030-27124-4.

[Fel23] Matt Feller. “Quasi-2-Segal sets”. English. In: *Tunis. J. Math.* 5.2 (2023), pp. 327–367. ISSN: 2576-7658. DOI: 10.2140/tunis.2023.5.327.

[GWB24] Daniel Gratzer, Jonathan Weinberger, and Ulrik Buchholtz. *Directed univalence in simplicial homotopy type theory*. 2024. arXiv: 2407.09146 [cs.LO]. URL: https://arxiv.org/abs/2407.09146.

[JT07] André Joyal and Myles Tierney. "Quasi-categories vs Segal spaces". In: *Categories in algebra, geometry and mathematical physics*. Vol. 431. Contemp. Math. Amer. Math. Soc., Providence, RI, 2007, pp. 277–326. ISBN: 978-0-8218-3970-6; 0-8218-3970-5. DOI: 10.1090/conm/431/08278. URL: https://doi.org/10.1090/conm/431/08278.

[RS17] Emily Riehl and Michael Shulman. "A type theory for synthetic ∞-categories". In: *Higher Structures* 1.1 (2017), pp. 147–224. URL: https://higher-structures.math.cas.cz/api/files/issues/Vol1Iss1/RiehlShulman.

[Shu15] Michael Shulman. "The univalence axiom for elegant Reedy presheaves". In: *Homology Homotopy Appl.* 17.2 (2015), pp. 81–106. ISSN: 1532-0073,1532-0081. DOI: 10.4310/HHA.2015.v17.n2.a6. URL: https://doi.org/10.4310/HHA.2015.v17.n2.a6.